

```
> restart;
```

```
> dp:=(f,g)->int(f*g,x=-1..1);
```

$$dp := (f, g) \mapsto \int_{-1}^1 fg \, dx \quad (1)$$

```
> nm:=f->sqrt(dp(f,f));
```

$$nm := f \mapsto \sqrt{dp(f, f)} \quad (2)$$

```
> dp(x^2,x^4);
```

$$\frac{2}{7} \quad (3)$$

```
> nm(x);
```

$$\frac{\sqrt{6}}{3} \quad (4)$$

```
> w[0]:=1;
```

$$w_0 := 1 \quad (5)$$

```
> v[0]:=w[0]/nm(w[0]);
```

$$v_0 := \frac{\sqrt{2}}{2} \quad (6)$$

```
> w[1]:=x-dp(v[0],x)*v[0];
```

$$w_1 := x \quad (7)$$

```
> v[1]:=w[1]/nm(w[1]);
```

$$v_1 := \frac{x\sqrt{6}}{2} \quad (8)$$

```
> w[2]:=x^2-dp(v[0],x^2)*v[0]-dp(v[1],x^2)*v[1];
```

$$w_2 := x^2 - \frac{1}{3} \quad (9)$$

```
> v[2]:=expand(w[2]/nm(w[2]),x);
```

$$v_2 := \frac{3\sqrt{10}}{4}x^2 - \frac{\sqrt{10}}{4} \quad (10)$$

```
> N:=8;
```

$$N := 8 \quad (11)$$

```
> for k from 0 to N do
  w[k]:=x^k;
  for j from 0 to k-1 do
    w[k]:=w[k]-dp(v[j],x^k)*v[j];
  end do;
  v[k]:=w[k]/nm(w[k]);
  print(k,v[k]);
end do;
```


$$0, \frac{\sqrt{2}}{2}$$

$$1, \frac{x\sqrt{6}}{2}$$

$$\begin{aligned}
& 2, \frac{3 \left(x^2 - \frac{1}{3} \right) \sqrt{10}}{4} \\
& 3, \frac{5 \left(x^3 - \frac{3}{5} x \right) \sqrt{14}}{4} \\
& 4, \frac{105 \left(x^4 + \frac{3}{35} - \frac{6}{7} x^2 \right) \sqrt{2}}{16} \\
& 5, \frac{63 \left(x^5 + \frac{5}{21} x - \frac{10}{9} x^3 \right) \sqrt{22}}{16} \\
& 6, \frac{231 \left(x^6 - \frac{5}{231} + \frac{5}{11} x^2 - \frac{15}{11} x^4 \right) \sqrt{26}}{32} \\
& 7, \frac{429 \left(x^7 - \frac{35}{429} x + \frac{105}{143} x^3 - \frac{21}{13} x^5 \right) \sqrt{30}}{32} \\
& 8, \frac{6435 \left(x^8 + \frac{7}{1287} - \frac{28}{143} x^2 + \frac{14}{13} x^4 - \frac{28}{15} x^6 \right) \sqrt{34}}{256}
\end{aligned}$$

(12)

```
> plot({v[5]},x=-1..1);
```



```
> R5:=solve(v[5]=0,x);
```

$$R5 := 0, \frac{\sqrt{245 - 14\sqrt{70}}}{21}, -\frac{\sqrt{245 - 14\sqrt{70}}}{21}, \frac{\sqrt{245 + 14\sqrt{70}}}{21}, -\frac{\sqrt{245 + 14\sqrt{70}}}{21} \quad (13)$$

```
> evalf(R5);
```

$$0., 0.5384693100, -0.5384693100, 0.9061798457, -0.9061798457 \quad (14)$$

```
> Digits:=16;
```

$$\text{Digits} := 16 \quad (15)$$

```
> evalf(R5);
```

$$0., 0.5384693101056829, -0.5384693101056829, 0.9061798459386638, -0.9061798459386638 \quad (16)$$